Code 83-24374-Opt4: 

“Apply a Route Layer's Definition Query to a Route Event layer”-ArcGIS 8.3 only
Public Sub CreateQueryLayer()

  On Error GoTo eh

  Dim pMxDoc As IMxDocument

  Dim i As Long

  Dim pMap As IMap

  Set pMxDoc = ThisDocument

  Set pMap = pMxDoc.FocusMap

  '+++ Get the route layer. It is called 'routes' in ArcMap's table of contents.

  Dim pLayer As ILayer

  Dim pFLayer As IFeatureLayer

  For i = 0 To pMap.LayerCount - 1

    Set pLayer = pMap.Layer(i)

    If LCase(pLayer.Name) = "routes" Then '+++ change this line to match your data

        If TypeOf pLayer Is IFeatureLayer Then

          Set pFLayer = pLayer

          Exit For

        End If

    End If

  Next i

  If pFLayer Is Nothing Then

    MsgBox "Could not find the route layer", vbExclamation, "CreateQueryLayer"

    Exit Sub

  End If

 '+++ Make sure the layer's feature class can support a query definition

 Dim pFC As IFeatureClass

 Dim pDS As IDataset

 Dim pWS As IWorkspace

 Set pFC = pFLayer.FeatureClass

 Set pDS = pFC

 Set pWS = pDS.Workspace

 If pWS.Type = esriFileSystemWorkspace Then

   MsgBox "You cannot create a query layer with this layer.", vbExclamation

   Exit Sub

 End If

 '+++ Get the route layer's definition query. We'll be using it below

  Dim pDef As IFeatureLayerDefinition

  Set pDef = pLayer

  Dim sDefExp As String

  sDefExp = pDef.DefinitionExpression

  If Not Len(sDefExp) > 0 Then

    'MsgBox "Route layer has no definition query.", vbExclamation

    Exit Sub

  End If

 '+++ Create a 'query' feature class. This is a virtual feature class that has

 '+++ less records than the original

  Dim pFWS As IFeatureWorkspace

  Dim pQueryDef As IQueryDef

  Dim pFeatureQueryName As IQueryName2

  Dim pWSN As IWorkspaceName

  Dim pTempDS As IDataset

  Dim pTempDSN As IDatasetName

  Dim pName As IName

  Set pFWS = pWS

  Set pQueryDef = pFWS.CreateQueryDef

  With pQueryDef

    ' .SubFields = '+++ optionally reduce the fields brought over

    .Tables = pDS.BrowseName

    .WhereClause = sDefExp

  End With

  Set pFeatureQueryName = New FeatureQueryName

  With pFeatureQueryName

    .PrimaryKey = pFC.OIDFieldName

    .QueryDef = pQueryDef

    .CopyLocally = False

  End With

  Set pTempDS = pFWS

  Set pWSN = pTempDS.FullName

  Set pTempDSN = pFeatureQueryName

  Set pTempDSN.WorkspaceName = pWSN

  pTempDSN.Name = pLayer.Name & "_query"

  Set pName = pFeatureQueryName

  '+++ Create a new feaure layer and set it's feature class to be the 'query' feature class

  Dim pNewLayer As IFeatureLayer

  Set pNewLayer = New FeatureLayer

  Set pNewLayer.FeatureClass = pName.Open

  pNewLayer.Name = pTempDSN.Name

  pMap.AddLayer pNewLayer

  pMxDoc.UpdateContents

  pMxDoc.ActiveView.Refresh

  Exit Sub

eh:

  MsgBox Err.Number & ": " & Err.Description, vbCritical, "CopyRouteLayerDefQuery"

End Sub

